JSAG **2** (2010), 20 – 23

The Journal of Software for
Algebra and Geometry

Simplicial Decomposability

DAVID COOK II

ABSTRACT. We introduce a new *Macaulay2* package, *SimplicialDecomposability*, which works in conjunction with the extant package *SimplicialComplexes* in order to compute a shelling order, if one exists, of a specified simplicial complex. Further, methods for determining vertex-decomposability are implemented, along with methods for determining k-decomposability.

INTRODUCTION. Given a finite vertex set V, a *simplicial complex* Δ is a set of subsets of V such that $\tau \in \Delta$ whenever $\tau \subset \sigma$ for some $\sigma \in \Delta$ and such that $\{v\} \in \Delta$ for all $v \in V$. The elements $\sigma \in \Delta$ are called *faces* or *simplices* and the maximal faces, i.e., those not contained in any other face, are called *facets*. The *dimension* of the face σ is $\#\sigma - 1$ and the *dimension* of Δ is $\max \dim \sigma$. Let $d = \dim \Delta + 1$. The f*-vector* of Δ is the $(d+1)$-tuple $(f_{-1}, \ldots, f_{d-1})$, where f_i is the number of faces of dimension i in Δ. Using this, the h*-vector* of Δ is the $(d+1)$-tuple $(h_0, \ldots, h_d)$ given by $h_j = \sum_{i=0}^{j} (-1)^{j-i} \binom{d-i}{j-i} f_{i-1}$ for $0 \le j \le d$.

Given a field K, let $K[V]$ be the polynomial ring with variables indexed by the vertices V. The *Stanley-Reisner ideal* of $\Delta$is the ideal $I(\Delta)$ in $K[V]$ generated by the minimal non-faces of Δ and the *Stanley-Reisner ring* of Δ is the ring $K[\Delta] = K[V]/I(\Delta)$. Thus the Stanley-Reisner ideals of complexes on a given vertex set V are exactly the squarefree monomial ideals in $K[V]$. Given the relations between the complex and the ideal, one can use tools from both algebra and combinatorics to study properties of both. For example, the h-vector of a complex Δ is the coefficient-vector of the numerator of the Hilbert series of $K[\Delta]$.

The package *SimplicialComplexes* by Sorin Popescu, Gregory G. Smith, and Mike Stillman already implements many methods for simplicial complexes in *Macaulay2* [M2], a software system designed to aid in research of commutative algebra and algebraic geometry. We introduce a new package, *SimplicialDecomposability*, for *Macaulay2* which provides several new methods for testing various forms of decomposability for simplicial complexes. Particularly, the package implements methods for testing shellability and vertex-decomposability.

SHELLABILITY. Given a finite set σ, let 2^σ be the set of all subsets of σ. Let Δ be a simplicial complex that has equidimensional facets, i.e., is *pure*. Then by Definition III.2.1 in [S], Δ is *shellable* if its facets can be ordered $\sigma_1, \ldots, \sigma_n$ so that $\bigcup_{j=1}^{i} 2^{\sigma_j} \setminus \bigcup_{j=1}^{i-1} 2^{\sigma_j}$ has a unique minimal element for $2 \le i \le n$, such an ordering is called a *shelling order*. See Definition 2.1 in [BW1] for the definition of non-pure shellability, which is implemented in the package for non-pure complexes.

Shellability is of interest because it implies a number of nice properties. In particular, if a pure simplicial complex is shellable, then its Stanley-Reisner ring is Cohen-Macaulay over every field [S, Theorem III.2.5]. Moreover, its h-vector is non-negative and can be read off from any shelling

2010 *Mathematics Subject Classification.* 05E45, 13F55.

SimplicialDecomposability version 1.0.5.

order [S, Theorem III.2.3]. Further still, the h-vectors of shellable pure complexes are numerically characterized [S, Theorems II.2.2 and II.3.3].

We recall that the *Alexander dual* of a simplicial complex Δ on vertex set V is the simplicial complex $\Delta^\vee := \{V \setminus F \mid F \notin \Delta\}$. Further, we say an ideal I has *linear quotients* if the minimal generators of I can be ordered $f_1, \ldots, f_n$ such that for $2 \leq i \leq n$, the quotient ideal $(f_1, \ldots, f_{i-1}) : (f_i)$ is generated by linear forms, in this case the sequence $\{(f_1) : (f_2), (f_1, f_2) : (f_3), \ldots, (f_1, \ldots, f_{n-1}) : f_n\}$ is called a *linear quotient order* of I with respect to $f_1, \ldots, f_n$.

In the following example we demonstrate Theorem 1.4(c) of [HHZ], which shows that a pure simplicial complex is shellable if and only if the Stanley-Reisner ideal of the Alexander dual has linear quotients. We begin by constructing the polynomial ring $R = \mathbb{Q}[a,b,c,d,e,f,g]$ and a simplicial complex D, which we verify is pure. Loading the package `SimplicialDecomposability` automatically loads the package `SimplicialComplexes`.

```
i1 : loadPackage "SimplicialDecomposability";
i2 : R = QQ[a..g];
i3 : D = simplicialComplex {c*e*g, b*e*g, a*e*g, b*d*g, a*d*g, c*e*f, b*e*f, a*e*f};
i4 : isPure D
o4 = true
```

We can recover a sequence of linear quotients directly from a shelling order. We recall that a pure simplicial complex Δ is shellable if there is an order of the facets $F_1, \ldots, F_n$ such that for $0 < j < i$ there exists an $x \in F_i \setminus F_j$ and a $0 < k < i$ such that $F_i \setminus F_k = \{x\}$. The set of vertices associated to each i in the preceding statement generate the linear quotient order of $I(\Delta^\vee)$ with respect to the given shelling order (see the proof of Theorem 1.4(c) in [HHZ]).

```
i5 : linearQuotients = O -> for i from 1 to #O-1 list (
          unique flatten for j from 0 to i-1 list (
               ImJ = set support O_i - set support O_j;
               for k from 0 to i - 1 list (
                    ImK = set support O_i - set support O_k;
                    if #ImK == 1 and isSubset(ImK, ImJ) then
                    first toList ImK else continue)));
```

We generate a shelling order O_1 of D with the method `shellingOrder`. This method attempts to build up a shelling order of D recursively using a depth-first search, adding one facet at a time. We note that in the non-pure case, the method only searches the remaining facets of largest dimension.

```
i6 : O1 = shellingOrder D
o6 = {c*e*g, b*e*g, a*e*g, b*d*g, a*d*g, c*e*f, b*e*f, a*e*f}
o6 : List
i7 : linearQuotients O1
o7 = {{b}, {a}, {d}, {d, a}, {f}, {f, b}, {f, a}}
o7 : List
```

It is sometimes beneficial to have more than one shelling order for a given simplicial complex. We can use the option `Random` with the method `shellingOrder` to first apply a random permutation to the facets before preceding with the recursion.

```
i8 : O2 = shellingOrder(D, Random => true)
o8 = {b*d*g, a*d*g, a*e*g, b*e*f, c*e*g, a*e*f, b*e*g, c*e*f}
```

```
o8 : List
i9 : linearQuotients O2
o9 = {{a}, {e}, {}, {c}, {f, a}, {e, b, g}, {c, f}}
o9 : List
```

Alternately, we may use the option `Permutation` with the method `shellingOrder` to force a given permutation on the facets before preceding with the recursion.

```
i10 : O3 = shellingOrder(D, Permutation => {3,2,1,0,4,5,6,7})
o10 = {b*d*g, b*e*g, a*e*g, c*e*g, a*d*g, c*e*f, b*e*f, a*e*f}
o10 : List
i11 : linearQuotients O3
o11 = {{e}, {a}, {c}, {a, d}, {f}, {f, b}, {f, a}}
o11 : List
```

Thus we now have multiple linear quotient orders associated to the ideal $I(D^\vee)$, each coming from a shelling order of D.

VERTEX-DECOMPOSABILITY. Let Δ be a pure simplicial complex and σ a face of Δ. Then the *link* and *face deletion* of σ in Δ are the simplicial complexes

$$\operatorname{link}_\Delta \sigma := \{\tau \in \Delta \mid \sigma \cap \tau = \emptyset, \sigma \cup \tau \in \Delta\} \text{ and } \operatorname{del}_\Delta \sigma := \{\tau \in \Delta \mid \sigma \nsubseteq \tau\}.$$

Definition 2.1 in [PB] defines Δ to be *vertex-decomposable* if either Δ is a simplex or there exists a vertex $x \in \Delta$, called a *shedding vertex*, such that $\operatorname{link}_\Delta x$ and $\operatorname{del}_\Delta x$ are vertex-decomposable.

See Definition 11.1 in [BW2] for the definition of non-pure vertex-decomposability, which is implemented in the package for non-pure complexes. Also, see Definitions 3.1 and 3.6 in [W] for the generalization of vertex-decomposability, called k-decomposability. It is implemented in the package with the methods `iskDecomposable` and `isSheddingFace`.

Being vertex-decomposable is a strong property which implies many things. A pure vertex-decomposable simplicial complex is shellable [PB, Theorem 2.8] and hence has non-negative h-vector [S, Theorem III.2.3] and its Stanley-Reisner ring is Cohen-Macaulay [S, Theorem III.2.5]. Furthermore, the h-vectors are numerically characterised for vertex-decomposable simplicial complexes [L, Theorem 3.5]. Moreover, the Stanley-Reisner ring of a pure vertex-decomposable complex is squarefree glicci [NR, Definition 2.2 and Theorem 3.3].

In the following example we demonstrate that the simplicial complex D from the previous example is indeed squarefree glicci. We use [NR, Remark 2.4] to find a basic double link of $I(D)$ to $I(\operatorname{link}_D v)$, both in R, for some shedding vertex v of D.

First, we verify that D is vertex-decomposable. Then we find its shedding vertices.

```
i12 : isVertexDecomposable D
o12 = true
i13 : select(allFaces(0, D), v -> isSheddingVertex(v, D))
o13 = {a, b, c, d, f}
o13 : List
```

We choose the shedding vertex f of D and generate $E = \operatorname{link}_D f$. Then we find its shedding vertices.

```
i14 : E = link(D, f);
i15 : ideal E
```

```
o15 = ideal (a*b, a*c, b*c, d, f, g)
o15 : Ideal of R
i16 : select(allFaces(0, E), v -> isSheddingVertex(v, E))
o16 = {a, b, c}
o16 : List
```

We now choose the shedding vertex c of E and generate $F = \operatorname{link}_E c$. Notice then that the Stanley-Reisner ideal of F is a complete intersection.

```
i17 : F = link(E, c);
i18 : ideal F
o18 = ideal (a, b, c, d, f, g)
o18 : Ideal of R
```

Hence, we now have the following sequence of basic double links in R which has squarefree terms on the even steps (the odd steps are omitted):

$$\mathbb{Q}[D] = (ab, ac, bc, cd, de, df, fg) \sim \mathbb{Q}[E] = (ab, ac, bc, d, f, g) \sim \mathbb{Q}[F] = (a, b, c, d, f, g).$$

ACKNOWLEDGEMENTS. The author would like to thank his advisor, Uwe Nagel, for reading drafts of this article and making comments thereon. The author would also like to thank Russ Woodroofe for pointing out the non-pure generalisation of k-decomposability in [W] and an anonymous referee for inspiring a more concrete example herein.

Part of the work for this paper was done while the author was at a *Macaulay2* workshop in Berkeley, California, January 8, 2010 through January 12, 2010, organised by Amelia Taylor and Hirotachi Abo with David Eisenbud, Daniel R. Grayson, and Michael E. Stillman, and funded by the National Security Agency (NSA) through grant H98230-09-1-0111.

REFERENCES.

[BW1] A. Björner and M.L. Wachs, *Shellable nonpure complexes and posets. I*, Trans. Amer. Math. Soc. **348** (1996), no. 4, 1299 – 1327.

[BW2] ______, *Shellable nonpure complexes and posets. II*, Trans. Amer. Math. Soc. **349** (1997), no. 10, 3945 – 3975.

[HHZ] J. Herzog, T. Hibi, and X. Zheng, *Dirac's theorem on chordal graphs and Alexander duality*, European J. Combin. **25** (2004), no. 7, 949 – 960.

[L] C.W. Lee, *Two combinatorial properties of a class of simplicial polytopes*, Israel J. Math. **47** (1984), no. 4, 261 – 269.

[M2] D.R. Grayson and M.E. Stillman, *Macaulay 2, a software system for research in algebraic geometry*, available at www.math.uiuc.edu/Macaulay2/.

[NR] U. Nagel and T. Römer, *Glicci simplicial complexes*, J. Pure Appl. Algebra **212** (2008), no. 10, 2250 – 2258.

[PB] J.S. Provan and L.J. Billera, *Decompositions of simplicial complexes related to diameters of convex polyhedra*, Math. Oper. Res. **5** (1980), no. 4, 576 – 594.

[S] R.P. Stanley, *Combinatorics and commutative algebra*, 2nd ed., Progress in Mathematics, vol. 41, Birkhäuser Boston Inc., Boston, MA, 1996.

[W] R. Woodroofe, *Chordal and sequentially Cohen-Macaulay clutters*, available at arXiv:0911.4697v2[math.CO].

RECEIVED : 2010-02-18 REVISED : 2010-06-10 ACCEPTED : 2010-08-03

dcook@ms.uky.edu : Department of Mathematics, University of Kentucky, 715 Patterson Office Tower, Lexington, KY 40506-0027, USA.